
mockgyver is an Erlang tool which will make it easier
to write EUnit tests which need to replace or alter
(stub/mock) the behaviour of other modules.
mockgyver aims to make that process as easy as possible
with a readable and concise syntax.
mockgyver is built around two main constructs:
?WHEN which makes it possible to alter the
behaviour of a function and another set of macros (like
?WAS_CALLED) which check that a function was called
with a chosen set of arguments.
Read more about constructs and syntax in the
documentation for the mockgyver module.
The documentation is generated using the edown
extension which generates documentation which is
immediately readable on github.  Remove the edown lines
from rebar.config to generate regular edoc.
A quick tutorial
Let's assume we want to make sure a fictional program
sets up an ssh connection correctly (in order to test
the part of our program which calls ssh:connect/3)
without having to start an ssh server.  Then we can use
the ?WHEN macro to replace the original ssh module and
let connect/3 return a bogus ssh_connection_ref():
    ?WHEN(ssh:connect(_Host, _Port, _Opts) -> {ok, ssh_ref}),
Also, let's mock close/1 while we're at it to make sure
it won't crash on the bogus ssh_ref:
    ?WHEN(ssh:close(_ConnRef) -> ok),
When testing our program, we want to make sure it calls
the ssh module with the correct arguments so we'll add
these lines:
    ?WAS_CALLED(ssh:connect({127,0,0,1}, 2022, [])),
    ?WAS_CALLED(ssh:close(ssh_ref)),
For all of this to work, the test needs to be
encapsulated within either the ?MOCK macro or the
?WITH_MOCKED_SETUP (recommended for eunit).  Assume the
test case above is within a function called
sets_up_and_tears_down_ssh_connection:
    sets_up_and_tears_down_ssh_connection_test() ->
        ?MOCK(fun sets_up_and_tears_down_ssh_connection/0).
Or, if you prefer ?WITH_MOCKED_SETUP:
    ssh_test_() ->
        ?WITH_MOCKED_SETUP(fun setup/0, fun cleanup/1).

    sets_up_and_tears_down_ssh_connection_test(_) ->
        ...
Sometimes a test requires a process to be started
before a test, and stopped after a test.  In that case,
the latter is better (it'll automatically export and
call all ...test/1 functions).
The final test case could look something like this:
    -include_lib("mockgyver/include/mockgyver.hrl").

    ssh_test_() ->
        ?WITH_MOCKED_SETUP(fun setup/0, fun cleanup/1).

    setup() ->
        ok.

    cleanup(_) ->
        ok.

    sets_up_and_tears_down_ssh_connection_test(_) ->
        ?WHEN(ssh:connect(_Host, _Port, _Opts) -> {ok, ssh_ref}),
        ?WHEN(ssh:close(_ConnRef) -> ok),
        ...start the program and trigger the ssh connection to open...
        ?WAS_CALLED(ssh:connect({127,0,0,1}, 2022, [])),
        ...trigger the ssh connection to close...
        ?WAS_CALLED(ssh:close(ssh_ref)),
API documentation
See mockgyver
for details.
Caveats
There are some pitfalls in using mockgyver that you
might want to know about.
	It's not possible to mock local functions.
This has to do with the way mockgyver works through
unloading and loading of modules.

	Take care when mocking modules which are used by
other parts of the system.
Examples include those in stdlib and kernel. A common
pitfall is mocking io. Since mockgyver is
potentially unloading and reloading the original
module many times during a test suite, processes
which are running that module may get killed as part
of the code loading mechanism within Erlang. A common
situation when mocking io is that eunit will stop
printing the progress and you will wonder what has
happened.

	NIFs cannot be mocked and mockgyver will try to
inform you if that is the case.

	mockgyver does currently not play well with cover and
cover will complain that a module has not been cover
compiled. This is probably solvable.


History
It all started when a friend of mine (Tomas
Abrahamsson) and I (Klas Johansson) wrote a tool we
called the stubber.  Using it looked something like this:
    stubber:run_with_replaced_modules(
        [{math, pi, fun() -> 4 end},
         {some_module, some_function, fun(X, Y) -> ... end},
         ...],
        fun() ->
            code which should be run with replacements above
        end),
Time went by and we had test cases which needed a more
intricate behaviour, the stubs grew more and more
complicated and that's when I thought: it must be
possible to come up with something better and that's
when I wrote mockgyver.
Using mockgyver with your own application
mockgyver is available as a hex package.  Just add it as a
dependency to your rebar.config:
{deps, [mockgyver]}.

... or with a specific version:
{deps, [{mockgyver, "some.good.version"}]}.

Building
Build mockgyver using rebar.  Also,
parse_trans (for the special syntax), edown
(for docs on github) and eunit_addons (for ?WITH_MOCKED_SETUP)
are required, but rebar takes care of that.
$ git clone git://github.com/klajo/mockgyver.git
$ rebar3 compile

Build docs using the edown library (for markdown format):
$ rebar3 as edown edoc

Build regular docs:
$ rebar3 edoc

Copyright (c) 2011, Klas Johansson
All rights reserved.

Redistribution and use in source and binary forms, with or without
modification, are permitted provided that the following conditions are
met:

    * Redistributions of source code must retain the above copyright
      notice, this list of conditions and the following disclaimer.

    * Redistributions in binary form must reproduce the above copyright
      notice, this list of conditions and the following disclaimer in
      the documentation and/or other materials provided with the
      distribution.

    * Neither the name of the copyright holder nor the names of its
      contributors may be used to endorse or promote products derived
      from this software without specific prior written permission.

THIS SOFTWARE IS PROVIDED BY THE COPYRIGHT HOLDERS AND CONTRIBUTORS "AS
IS" AND ANY EXPRESS OR IMPLIED WARRANTIES, INCLUDING, BUT NOT LIMITED
TO, THE IMPLIED WARRANTIES OF MERCHANTABILITY AND FITNESS FOR A
PARTICULAR PURPOSE ARE DISCLAIMED. IN NO EVENT SHALL THE COPYRIGHT
HOLDER OR CONTRIBUTORS BE LIABLE FOR ANY DIRECT, INDIRECT, INCIDENTAL,
SPECIAL, EXEMPLARY, OR CONSEQUENTIAL DAMAGES (INCLUDING, BUT NOT LIMITED
TO, PROCUREMENT OF SUBSTITUTE GOODS OR SERVICES; LOSS OF USE, DATA, OR
PROFITS; OR BUSINESS INTERRUPTION) HOWEVER CAUSED AND ON ANY THEORY OF
LIABILITY, WHETHER IN CONTRACT, STRICT LIABILITY, OR TORT (INCLUDING
NEGLIGENCE OR OTHERWISE) ARISING IN ANY WAY OUT OF THE USE OF THIS
SOFTWARE, EVEN IF ADVISED OF THE POSSIBILITY OF SUCH DAMAGE.

Mock functions and modules
[bookmark: Initiating_mock]Initiating mock
In order to use the various macros below, mocking must be initiated using the ?MOCK macro or ?WITH_MOCKED_SETUP (recommended from eunit tests).
[bookmark: ?MOCK_syntax]?MOCK syntax
       ?MOCK(Expr)
   
where Expr in a single expression, like a fun. The rest of the macros in this module can be used within this fun or in a function called by the fun.
[bookmark: ?WITH_MOCKED_SETUP_syntax]?WITH_MOCKED_SETUP syntax
       ?WITH_MOCKED_SETUP(SetupFun, CleanupFun),
       ?WITH_MOCKED_SETUP(SetupFun, CleanupFun, ForAllTimeout, PerTcTimeout),
       ?WITH_MOCKED_SETUP(SetupFun, CleanupFun, ForAllTimeout, PerTcTimeout,
                          Tests),
   
This is an easy way of using mocks from within eunit tests and is mock-specific version of the ?WITH_SETUP macro. See the docs for the ?WITH_SETUP macro in the eunit_addons project for more information on parameters and settings.
[bookmark: Mocking_a_function]Mocking a function
[bookmark: Introduction]Introduction
By mocking a function, its original side-effects and return value (or throw/exit/error) are overridden and replaced. This can be used to:
	replace existing functions in existing modules
	add new functions to existing modules
	add new modules

BIFs (built-in functions) cannot be mocked.
The original module will be renamed (a "^" will be appended to the original module name, i.e. foo will be renamed to 'foo^'). A mock can then call the original function just by performing a regular function call.
Since WHEN is a macro, and macros don't support argument lists (something like "Arg..."), multi-expression mocks must be surrounded by begin ... end to be treated as one argument by the preprocessor.
A mock that was introduced using the ?WHEN macro can be forgotten, i.e. returned to the behaviour of the original module, using the ?FORGET_WHEN macro.
[bookmark: ?WHEN_syntax]?WHEN syntax
       ?WHEN(module:function(Arg1, Arg2, ...) -> Expr),
   
where Expr is a single expression (like a term) or a series of expressions surrounded by begin and end.
[bookmark: ?FORGET_WHEN_syntax]?FORGET_WHEN syntax
       ?FORGET_WHEN(module:function(_, _, ...)),
   
The only things of interest are the name of the module, the name of the function and the arity. The arguments of the function are ignored and it can be a wise idea to set these to the "don't care" variable: underscore.
[bookmark: Examples]Examples
Note: Apparently the Erlang/OTP team doesn't want us to redefine PI to 4 anymore :-), since starting at R15B, math:pi/0 is marked as pure which means that the compiler is allowed to replace the math:pi() function call by a constant: 3.14... This means that even though mockgyver can mock the pi/0 function, a test case will never call math:pi/0 since it will be inlined. See commit 5adf009cb09295893e6bb01b4666a569590e0f19 (compiler: Turn calls to math:pi/0 into constant values) in the otp sources.
Redefine pi to 4:       ?WHEN(math:pi() -> 4),
   
Implement a mock with multiple clauses:       ?WHEN(my_module:classify_number(N) when N >= 0 -> positive;
             my_module:classify_number(_N)            -> negative),
   
Call original module:       ?WHEN(math:pi() -> 'math^':pi() * 2),
   
Use a variable bound outside the mock:       Answer = 42,
       ?WHEN(math:pi() -> Answer),
   
Redefine the mock:       ?WHEN(math:pi() -> 4),
       4 = math:pi(),
       ?WHEN(math:pi() -> 5),
       5 = math:pi(),
   
Let the mock exit with an error:       ?WHEN(math:pi() -> erlang:error(some_error)),
   
Make a new module:       ?WHEN(my_math:pi() -> 4),
       ?WHEN(my_math:e() -> 3),
   
Put multiple clauses in a function's body:       ?WHEN(math:pi() ->
                 begin
                     do_something1(),
                     do_something2()
                 end),
   
Revert the pi function to its default behaviour (return value from the original module), any other mocks in the same module, or any other module are left untouched:       ?WHEN(math:pi() -> 4),
       4 = math:pi(),
       ?FORGET_WHEN(math:pi()),
       3.1415... = math:pi(),
   
[bookmark: Validating_calls]Validating calls
[bookmark: Introduction]Introduction
There are a number of ways to check that a certain function has been called and that works for both mocks and non-mocks.
	?WAS_CALLED: Check that a function was called with certain set of parameters a chosen number of times. The validation is done at the place of the macro, consider this when verifying asynchronous procedures (see also ?WAIT_CALLED). Return a list of argument lists, one argument list for each call to the function. An argument list contains the arguments of a specific call. Will crash with an error if the criteria isn't fulfilled.
	?WAIT_CALLED: Same as ?WAS_CALLED, with a twist: waits for the criteria to be fulfilled which can be useful for asynchronous procedures.
	?GET_CALLS: Return a list of argument lists (just like ?WAS_CALLED or ?WAIT_CALLED) without checking any criteria.
	?NUM_CALLS: Return the number of calls to a function.
	?FORGET_CALLS: Forget the calls that have been logged. This exists in two versions:	One which forgets calls to a certain function. Takes arguments and guards into account, i.e. only the calls which match the module name, function name and all arguments as well as any guards will be forgotten, while the rest of the calls remain.
	One which forgets all calls to any function.



[bookmark: ?WAS_CALLED_syntax]?WAS_CALLED syntax
       ?WAS_CALLED(module:function(Arg1, Arg2, ...)),
           equivalent to ?WAS_CALLED(module:function(Arg1, Arg2, ...), once)
       ?WAS_CALLED(module:function(Arg1, Arg2, ...), Criteria),
           Criteria = once | never | {times, N} | {at_least, N} | {at_most, N}
           N = integer()
  
           Result: [CallArgs]
                   CallArgs = [CallArg]
                   CallArg = term()
  
   
[bookmark: ?WAIT_CALLED_syntax]?WAIT_CALLED syntax
See syntax for ?WAS_CALLED.
[bookmark: ?GET_CALLS_syntax]?GET_CALLS syntax
       ?GET_CALLS(module:function(Arg1, Arg2, ...)),
  
           Result: [CallArgs]
                   CallArgs = [CallArg]
                   CallArg = term()
   
[bookmark: ?NUM_CALLS_syntax]?NUM_CALLS syntax
       ?NUM_CALLS(module:function(Arg1, Arg2, ...)),
  
           Result: integer()
   
[bookmark: ?FORGET_CALLS_syntax]?FORGET_CALLS syntax
       ?FORGET_CALLS(module:function(Arg1, Arg2, ...)),
       ?FORGET_CALLS(),
   
[bookmark: Examples]Examples
Check that a function has been called once (the two alternatives are equivalent):       ?WAS_CALLED(math:pi()),
       ?WAS_CALLED(math:pi(), once),
   
Check that a function has never been called:       ?WAS_CALLED(math:pi(), never),
   
Check that a function has been called twice:       ?WAS_CALLED(math:pi(), {times, 2}),
   
Check that a function has been called at least twice:       ?WAS_CALLED(math:pi(), {at_least, 2}),
   
Check that a function has been called at most twice:       ?WAS_CALLED(math:pi(), {at_most, 2}),
   
Use pattern matching to check that a function was called with certain arguments:       ?WAS_CALLED(lists:reverse([a, b, c])),
   
Pattern matching can even use bound variables:       L = [a, b, c],
       ?WAS_CALLED(lists:reverse(L)),
   
Use a guard to validate the parameters in a call:       ?WAS_CALLED(lists:reverse(L) when is_list(L)),
   
Retrieve the arguments in a call while verifying the number of calls:       a = lists:nth(1, [a, b]),
       d = lists:nth(2, [c, d]),
       [[1, [a, b]], [2, [c, d]]] = ?WAS_CALLED(lists:nth(_, _), {times, 2}),
   
Retrieve the arguments in a call without verifying the number of calls:       a = lists:nth(1, [a, b]),
       d = lists:nth(2, [c, d]),
       [[1, [a, b]], [2, [c, d]]] = ?GET_CALLS(lists:nth(_, _)),
   
Retrieve the number of calls:       a = lists:nth(1, [a, b]),
       d = lists:nth(2, [c, d]),
       2 = ?NUM_CALLS(lists:nth(_, _)),
   
Forget calls to functions:       a = lists:nth(1, [a, b, c]),
       e = lists:nth(2, [d, e, f]),
       i = lists:nth(3, [g, h, i]),
       ?WAS_CALLED(lists:nth(1, [a, b, c]), once),
       ?WAS_CALLED(lists:nth(2, [d, e, f]), once),
       ?WAS_CALLED(lists:nth(3, [g, h, i]), once),
       ?FORGET_CALLS(lists:nth(2, [d, e, f])),
       ?WAS_CALLED(lists:nth(1, [a, b, c]), once),
       ?WAS_CALLED(lists:nth(2, [d, e, f]), never),
       ?WAS_CALLED(lists:nth(3, [g, h, i]), once),
       ?FORGET_CALLS(lists:nth(_, _)),
       ?WAS_CALLED(lists:nth(1, [a, b, c]), never),
       ?WAS_CALLED(lists:nth(2, [d, e, f]), never),
       ?WAS_CALLED(lists:nth(3, [g, h, i]), never),
   
Forget calls to all functions:       a = lists:nth(1, [a, b, c]),
       e = lists:nth(2, [d, e, f]),
       i = lists:nth(3, [g, h, i]),
       ?WAS_CALLED(lists:nth(1, [a, b, c]), once),
       ?WAS_CALLED(lists:nth(2, [d, e, f]), once),
       ?WAS_CALLED(lists:nth(3, [g, h, i]), once),
       ?FORGET_CALLS(),
       ?WAS_CALLED(lists:nth(1, [a, b, c]), never),
       ?WAS_CALLED(lists:nth(2, [d, e, f]), never),
       ?WAS_CALLED(lists:nth(3, [g, h, i]), never),
   
%%% ==== ?MOCK_SESSION_PARAMS ====
This is expands to a term that describes MFAs that are (to be) mocked (with ?WHEN) and to be watched or traced (with ?WAS_CALLED and similar). It can be used with the start_session or exec functions.

          -type args() :: [term()].

          -type args_expectation() :: function().

          -type checksum() :: term().

          -type criteria() :: once | {at_least, integer()} | {at_most, integer()} | {times, integer()} | never.

          -type mf_args_expectation() :: {module(), atom(), args_expectation()}.

          -type session_params() :: {Mocked :: [mfa()], Watched :: [mfa()]}.

          -type state_name() :: no_session | session.

          -type verify_op() ::
    {was_called, criteria()} |
    {wait_called, criteria()} |
    num_calls | get_calls | forget_when | forget_calls.

          -type verify_opt() :: {location, {File :: string(), Line :: integer()}}.

          -spec end_session() -> ok.

          -spec start_session(session_params()) -> ok | {error, Reason :: term()}.

